
5 First-Order Methods

The previous chapter introduced the general concept of descent directionmethods.
This chapter discusses a variety of algorithms that use first-order methods to
select the appropriate descent direction. First-order methods rely on gradient
information to help direct the search for a minimum, which can be obtained using
methods outlined in chapter 2.

5.1 Gradient Descent

An intuitive choice for descent direction d is the direction of steepest descent.
Following the direction of steepest descent is guaranteed to lead to improvement,
provided that the objective function is smooth, the step size is sufficiently small,
and we are not already at a point where the gradient is zero.1 The direction of 1 Apoint where the gradient is zero

is called a stationary point.steepest descent is the direction opposite the gradient∇ f , hence the name gradient
descent. For convenience in this chapter, we define

g(k) = ∇ f (x(k)) (5.1)

where x(k) is our design point at descent iteration k.
In gradient descent, we typically normalize the direction of steepest descent

(see example 5.1):
d(k) = − g(k)

‖g(k)‖ (5.2)

Jagged search paths result if we choose a step size that leads to the maximal
decrease in f . In fact, the next direction will always be orthogonal to the current
direction. We can show this as follows:

Suppose we have f (x) = x1x2
2. The gradient is ∇ f = [x2

2, 2x1x2]. For x(k) =

[1, 2] we get an unnormalized direction of steepest descent d = [−4,−4],
which is normalized to d = [− 1√

2
,− 1√

2
].

Example 5.1. Computing the gradi-
ent descent direction.

If we optimize the step size at each step, we have

α(k) = arg min
α

f (x(k) + αd(k)) (5.3)

The optimization above implies that the directional derivative equals zero. Using
equation (2.9), we have

∇ f (x(k) + αd(k))⊤d(k) = 0 (5.4)

We know
d(k+1) = − ∇ f (x(k) + αd(k))

‖∇ f (x(k) + αd(k))‖ (5.5)

Hence,
d(k+1)⊤d(k) = 0 (5.6)

which means that d(k+1) and d(k) are orthogonal.
Narrow valleys aligned with a descent direction are not an issue. When the

descent directions cross over the valley, many steps must be taken in order to
make progress along the valley’s floor as shown in figure 5.1. An implementation
of gradient descent is provided by algorithm 5.1.

5.2 Conjugate Gradient

Gradient descent can perform poorly in narrow valleys. The conjugate gradient
method overcomes this issue by borrowing inspiration from methods for opti-
mizing quadratic functions:

minimize
x

f (x) =
1

2
x⊤Ax + b⊤x + c (5.7)

where A is symmetric and positive definite, and thus f has a unique local mini-
mum (section 1.6.2).

© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



5.2. conjugate gradient 71

Figure 5.1. Gradient descent can
result in zig-zagging in narrow
canyons. Here we see the effect
on the Rosenbrock function (ap-
pendix B.6).

abstract type DescentMethod end
struct GradientDescent <: DescentMethod

α
end
init!(M::GradientDescent, f, ∇f, x) = M
function step!(M::GradientDescent, f, ∇f, x)

α, g = M.α, ∇f(x)
return x - α*g

end

Algorithm 5.1. The gradient de-
scent method, which follows the
direction of gradient descent with
a fixed learning rate. The step!
function produces the next iterate
whereas the init function does
nothing.

The conjugate gradient method can optimize n-dimensional quadratic func-
tions in n steps as shown in figure 5.2. Its directions are mutually conjugate with
respect to A:

d(i)⊤A d(j) = 0 for all i 6= j (5.8)
The mutually conjugate vectors are the basis vectors of A. They are generally not
orthogonal to one another.

1
2

3

x1

x
2

Figure 5.2. Conjugate gradient
descent converges in n steps
when applied to an n-dimensional
quadratic function.

The successive conjugate directions are computed using gradient information
and the previous descent direction. The algorithm starts with the direction of
steepest descent:

d(1) = −g(1) (5.9)
We then use line search to find the next design point. For quadratic functions, the
step factor α can be computed exactly (example 5.2). The update is then:

x(2) = x(1) + α(1)d(1) (5.10)

Suppose we want to derive the optimal step factor for a line search on a
quadratic function:

minimize
α

f (x + αd)

We can compute the derivative with respect to α:

∂ f (x + αd)

∂α
=

∂

∂α

[
1

2
(x + αd)⊤A(x + αd) + b⊤(x + αd) + c

]

= d⊤A(x + αd) + d⊤b

= d⊤(Ax + b) + αd⊤Ad

Setting ∂ f (x+αd)
∂α = 0 results in:

α = −d⊤(Ax + b)

d⊤Ad

Example 5.2. The optimal step fac-
tor for a line search on a quadratic
function.

Subsequent iterations choose d(k+1) based on the current gradient and a con-
tribution from the previous descent direction:

d(k) = −g(k) + β(k)d(k−1) (5.11)

for scalar parameter β. Larger values of β indicate that the previous descent
direction contributes more strongly.

We can derive the best value for β for a known A, using the fact that d(k) is
conjugate to d(k−1):

d(k)⊤Ad(k−1) = 0 (5.12)
⇒ (−g(k) + β(k)d(k−1))⊤Ad(k−1) = 0 (5.13)

⇒ −g(k)⊤Ad(k−1) + β(k)d(k−1)⊤Ad(k−1) = 0 (5.14)

⇒ β(k) =
g(k)⊤Ad(k−1)

d(k−1)⊤Ad(k−1)
(5.15)

The conjugate gradient method can be applied to nonquadratic functions as
well. Smooth, continuous functions behave like quadratic functions close to a
local minimum, and the conjugate gradient method will converge very quickly in
such regions.

Unfortunately, we do not know the value of A that best approximates f around
x(k). Instead, several choices for β(k) tend to work well:

Fletcher-Reeves:2 2 R. Fletcher and C.M. Reeves,
“Function Minimization by Conju-
gate Gradients,” The Computer Jour-
nal, vol. 7, no. 2, pp. 149–154, 1964.

β(k) =
g(k)⊤g(k)

g(k−1)⊤g(k−1)
(5.16)

Polak-Ribière:3 3 E. Polak and G. Ribière, “Note
sur la Convergence de Méthodes
de Directions Conjuguées,”
Revue Française d’informatique et
de Recherche Opérationnelle, Série
Rouge, vol. 3, no. 1, pp. 35–43, 1969.

β(k) =
g(k)⊤

(

g(k) − g(k−1)
)

g(k−1)⊤g(k−1)
(5.17)

Convergence for the Polak-Ribière method (algorithm 5.2) can be guaranteed
if we modify it to allow for automatic resets:

β← max(β, 0) (5.18)

Figure 5.3 shows an example search using this method.

Figure 5.3. The conjugate gradient
method with the Polak-Ribière up-
date. Gradient descent is shown in
gray.

mutable struct ConjugateGradientDescent <: DescentMethod
d
g

end
function init!(M::ConjugateGradientDescent, f, ∇f, x)

M.g = ∇f(x)
M.d = -M.g
return M

end
function step!(M::ConjugateGradientDescent, f, ∇f, x)

d, g = M.d, M.g
g′ = ∇f(x)
β = max(0, dot(g′, g′-g)/(g⋅g))
d′ = -g′ + β*d
x′ = line_search(f, x, d′)
M.d, M.g = d′, g′
return x′

end

Algorithm 5.2. The conjugate gradi-
ent method with the Polak-Ribière
update, where d is the previous
search direction and g is the pre-
vious gradient.

5.3 Momentum

Figure 5.4. Regions that are nearly
flat have gradients with small mag-
nitudes and can thus require many
iterations of gradient descent to tra-
verse.

Gradient descent will take a long time to traverse a nearly flat surface as shown
in figure 5.4. Allowing momentum to accumulate is one way to speed progress.
We can modify gradient descent to incorporate momentum.

The momentum update equations are:

v(k+1) = βv(k) − αg(k) (5.19)
x(k+1) = x(k) + v(k+1) (5.20)

For β = 0, we recover gradient descent. Momentum can be interpreted as a ball
rolling down a nearly horizontal incline. The ball naturally gathers momentum
as gravity causes it to accelerate, just as the gradient causes momentum to accu-
mulate in this descent method. An implementation is provided in algorithm 5.3.
Momentum descent is compared to gradient descent in figure 5.5.

mutable struct Momentum <: DescentMethod
α # learning rate
β # momentum decay
v # momentum

end
function init!(M::Momentum, f, ∇f, x)

M.v = zeros(length(x))
return M

end
function step!(M::Momentum, f, ∇f, x)

α, β, v, g = M.α, M.β, M.v, ∇f(x)
v[:] = β*v - α*g
return x + v

end

Algorithm 5.3. The momentum
method for accelerated descent.

The first line in step! makes
copies of the scalars α and β,
but creates a reference to the vec-
tor v. Thus, the following line
v[:] = β*v - α*g modifies the
original momentum vector in the
struct M.

© 2019 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2022-05-22 00:25:57-07:00, revision 47fd495, comments to bugs@algorithmsbook.com



76 chapter 5. f irst-order methods

x1

x
2

gradient descent
momentum

Figure 5.5. Gradient descent and
the momentum method compared
on the Rosenbrock function with
b = 100; see appendix B.6.

5.4 Nesterov Momentum

One issue of momentum is that the steps do not slow down enough at the bottom
of a valley and tend to overshoot the valley floor. Nesterov momentum4 modifies 4 Y.Nesterov, “AMethod of Solving

a Convex Programming Problem
with Convergence Rate O(1/k2),”
Soviet Mathematics Doklady, vol. 27,
no. 2, pp. 543–547, 1983.

the momentum algorithm to use the gradient at the projected future position:

v(k+1) = βv(k) − α∇ f (x(k) + βv(k)) (5.21)
x(k+1) = x(k) + v(k+1) (5.22)

An implementation is provided by algorithm 5.4. The Nesterov momentum and
momentum descent methods are compared in figure 5.6.

mutable struct NesterovMomentum <: DescentMethod
α # learning rate
β # momentum decay
v # momentum

end
function init!(M::NesterovMomentum, f, ∇f, x)

M.v = zeros(length(x))
return M

end
function step!(M::NesterovMomentum, f, ∇f, x)

α, β, v = M.α, M.β, M.v
v[:] = β*v - α*∇f(x + β*v)
return x + v

end

Algorithm 5.4. Nesterov’s momen-
tum method of accelerated de-
scent.

Figure 5.6. The momentum and
Nesterov momentum methods
compared on the Rosenbrock
function with b = 100; see
appendix B.6.

5.5 Adagrad

Momentum and Nesterov momentum update all components of x with the same
learning rate. The adaptive subgradient method, or Adagrad,5 adapts a learning 5 J. Duchi, E. Hazan, and Y. Singer,

“Adaptive Subgradient Methods
for Online Learning and Stochastic
Optimization,” Journal of Machine
Learning Research, vol. 12, pp. 2121–
2159, 2011.

rate for each component of x. Adagrad dulls the influence of parameters with
consistently high gradients, thereby increasing the influence of parameters with
infrequent updates.6

6 Adagrad excels when the gradi-
ent is sparse. The original paper
was motivated by stochastic gradi-
ent descent, which picks a random
batch of training data for each iter-
ation from which to compute a gra-
dient. Many deep learning datasets
for real-world problems produce
sparse gradients where some fea-
tures occur far less frequently than
others.

The Adagrad update step is:

x
(k+1)
i = x

(k)
i −

α

ǫ +
√

s
(k)
i

g
(k)
i (5.23)

where s(k) is a vector whose ith entry is the sum of the squares of the partials,
with respect to xi, up to time step k,

s
(k)
i =

k

∑
j=1

(

g
(j)
i

)2
(5.24)

and ǫ is a small value, on the order of 1× 10−8, to prevent division by zero.
Adagrad is far less sensitive to the learning rate parameter α. The learning rate

parameter is typically set to a default value of 0.01. Adagrad’s primary weakness
is that the components of s are each strictly nondecreasing. The accumulated sum
causes the effective learning rate to decrease during training, often becoming
infinitesimally small before convergence. An implementation is provided by
algorithm 5.5.

mutable struct Adagrad <: DescentMethod
α # learning rate
ϵ # small value
s # sum of squared gradient

end
function init!(M::Adagrad, f, ∇f, x)

M.s = zeros(length(x))
return M

end
function step!(M::Adagrad, f, ∇f, x)

α, ϵ, s, g = M.α, M.ϵ, M.s, ∇f(x)
s[:] += g.*g
return x - α*g ./ (sqrt.(s) .+ ϵ)

end

Algorithm 5.5. The Adagrad accel-
erated descent method.

5.6 RMSProp

RMSProp7 extends Adagrad to avoid the effect of a monotonically decreasing 7 RMSProp is unpublished and
comes from Lecture 6e of Geoff
Hinton’s Coursera class.learning rate. RMSProp maintains a decaying average of squared gradients. This

average is updated according to:8 8 The operation a ⊙ b is the
element-wise product between
vectors a and b.ŝ(k+1) = γŝ(k) + (1− γ)

(

g(k) ⊙ g(k)
)

(5.25)

where the decay γ ∈ [0, 1] is typically close to 0.9.
The decaying average of past squared gradients can be substituted into RM-

SProp’s update equation:9 9 The denominator is similar to the
root mean square (RMS) of the
gradient component. In this chap-
ter we use RMS(x) to refer to the
decaying root mean square of the
time series of x.

x
(k+1)
i = x

(k)
i −

α

ǫ +
√

ŝ
(k+1)
i

g
(k)
i (5.26)

= x
(k)
i −

α

ǫ + RMS(gi)
g
(k)
i (5.27)

An implementation is provided by algorithm 5.6.

5.7 Adadelta

Adadelta10 is another method for overcoming Adagrad’s monotonically decreasing 10 M.D. Zeiler, “ADADELTA: An
Adaptive Learning Rate Method,”
ArXiv, no. 1212.5701, 2012.learning rate. After independently deriving the RMSProp update, the authors

noticed that the units in the update equations for gradient descent, momentum,

mutable struct RMSProp <: DescentMethod
α # learning rate
γ # decay
ϵ # small value
s # sum of squared gradient

end
function init!(M::RMSProp, f, ∇f, x)

M.s = zeros(length(x))
return M

end
function step!(M::RMSProp, f, ∇f, x)

α, γ, ϵ, s, g = M.α, M.γ, M.ϵ, M.s, ∇f(x)
s[:] = γ*s + (1-γ)*(g.*g)
return x - α*g ./ (sqrt.(s) .+ ϵ)

end

Algorithm 5.6. The RMSProp accel-
erated descent method.

andAdagrad do not match. To fix this, they use an exponentially decaying average
of the square updates:

x
(k+1)
i = x

(k)
i −

RMS(∆xi)

ǫ + RMS(gi)
g
(k)
i (5.28)

which eliminates the learning rate parameter entirely. An implementation is
provided by algorithm 5.7.

5.8 Adam

The adaptive moment estimation method, or Adam,11 also adapts learning rates to 11 D. Kingma and J. Ba, “Adam: A
Method for Stochastic Optimiza-
tion,” in International Conference
on Learning Representations (ICLR),
2015.

each parameter (algorithm 5.8). It stores both an exponentially decaying squared
gradient like RMSProp and Adadelta, but also an exponentially decaying gradient
like momentum.

Initializing the gradient and squared gradient to zero introduces a bias. A bias
correction step helps alleviate the issue.12 The equations applied during each 12 According to the original

paper, good default settings are
α = 0.001, γv = 0.9, γs = 0.999,
and ǫ = 1× 10−8.

mutable struct Adadelta <: DescentMethod
γs # gradient decay
γx # update decay
ϵ # small value
s # sum of squared gradients
u # sum of squared updates

end
function init!(M::Adadelta, f, ∇f, x)

M.s = zeros(length(x))
M.u = zeros(length(x))
return M

end
function step!(M::Adadelta, f, ∇f, x)

γs, γx, ϵ, s, u, g = M.γs, M.γx, M.ϵ, M.s, M.u, ∇f(x)
s[:] = γs*s + (1-γs)*g.*g
Δx = - (sqrt.(u) .+ ϵ) ./ (sqrt.(s) .+ ϵ) .* g
u[:] = γx*u + (1-γx)*Δx.*Δx
return x + Δx

end

Algorithm 5.7. The Adadelta accel-
erated descent method. The small
constant ϵ is added to the numer-
ator as well to prevent progress
from entirely decaying to zero and
to start off the first iteration where
∆x = 0.

iteration for Adam are:

biased decaying momentum: v(k+1) = γvv(k) + (1− γv)g
(k) (5.29)

biased decaying sq. gradient: s(k+1) = γss
(k) + (1− γs)

(

g(k) ⊙ g(k)
)

(5.30)

corrected decaying momentum: v̂(k+1) = v(k+1)/(1− γk
v) (5.31)

corrected decaying sq. gradient: ŝ(k+1) = s(k+1)/(1− γk
s ) (5.32)

next iterate: x(k+1) = x(k) − αv̂(k+1)/
(

ǫ +
√

ŝ(k+1)
)

(5.33)

5.9 Hypergradient Descent

The accelerated descent methods are either extremely sensitive to the learning rate
or go to great lengths to adapt the learning rate during execution. The learning
rate dictates how sensitive the method is to the gradient signal. A rate that is too
high or too low often drastically affects performance.

mutable struct Adam <: DescentMethod
α # learning rate
γv # decay
γs # decay
ϵ # small value
k # step counter
v # 1st moment estimate
s # 2nd moment estimate

end
function init!(M::Adam, f, ∇f, x)

M.k = 0
M.v = zeros(length(x))
M.s = zeros(length(x))
return M

end
function step!(M::Adam, f, ∇f, x)

α, γv, γs, ϵ, k = M.α, M.γv, M.γs, M.ϵ, M.k
s, v, g = M.s, M.v, ∇f(x)
v[:] = γv*v + (1-γv)*g
s[:] = γs*s + (1-γs)*g.*g
M.k = k += 1
v_hat = v ./ (1 - γv^k)
s_hat = s ./ (1 - γs^k)
return x - α*v_hat ./ (sqrt.(s_hat) .+ ϵ)

end

Algorithm 5.8. The Adam acceler-
ated descent method.

Hypergradient descent13 was developed with the understanding that the deriva- 13 A.G. Baydin, R. Cornish, D.M.
Rubio, M. Schmidt, and F. Wood,
“Online Learning Rate Adaptation
with Hypergradient Descent,” in
International Conference on Learning
Representations (ICLR), 2018.

tive of the learning rate should be useful for improving optimizer performance. A
hypergradient is a derivative taken with respect to a hyperparameter. Hypergradi-
ent algorithms reduce the sensitivity to the hyperparameter, allowing it to adapt
more quickly.

Hypergradient descent applies gradient descent to the learning rate of an
underlying descent method. The method requires the partial derivative of the
objective function with respect to the learning rate. For gradient descent, this
partial derivative is:

∂ f (x(k+1))

∂α(k)
= (g(k+1))⊤

∂

∂α(k)

(

x(k) − α(k)g(k)
)

(5.34)

= (g(k+1))⊤
(

−g(k)
)

(5.35)

Computing the hypergradient thus requires keeping track of the last gradient.
The resulting update rule is:

α(k) = α(k−1) − µ
∂ f (x(k))

∂α(k−1)
(5.36)

= α(k−1) + µ(g(k))⊤g(k−1) (5.37)

where µ is the hypergradient learning rate.
This derivation can be applied to any gradient-based descent method that fol-

lows equation (4.1). Implementations are provided for the hypergradient versions
of gradient descent (algorithm 5.9) and Nesterov momentum (algorithm 5.10).
These methods are visualized in figure 5.7.

mutable struct HyperGradientDescent <: DescentMethod
α0 # initial learning rate
μ # learning rate of the learning rate
α # current learning rate
g_prev # previous gradient

end
function init!(M::HyperGradientDescent, f, ∇f, x)

M.α = M.α0
M.g_prev = zeros(length(x))
return M

end
function step!(M::HyperGradientDescent, f, ∇f, x)

α, μ, g, g_prev = M.α, M.μ, ∇f(x), M.g_prev
α = α + μ*(g⋅g_prev)
M.g_prev, M.α = g, α
return x - α*g

end

Algorithm 5.9. The hypergradient
form of gradient descent.

mutable struct HyperNesterovMomentum <: DescentMethod
α0 # initial learning rate
μ # learning rate of the learning rate
β # momentum decay
v # momentum
α # current learning rate
h_prev # previous gradient

end
function init!(M::HyperNesterovMomentum, f, ∇f, x)

M.α = M.α0
M.v = zeros(length(x))
M.h_prev = zeros(length(x))
return M

end
function step!(M::HyperNesterovMomentum, f, ∇f, x)

α, β, μ, v = M.α, M.β, M.μ, M.v
h, h_prev = ∇f(x + β*v), M.h_prev
α = α + μ*(h⋅h_prev)
v[:] = β*v - α*h
M.h_prev, M.α = h, α
return x + v

end

Algorithm 5.10. The hypergradient
form of the Nesterov momentum
descent method.

Figure 5.7. Hypergradient ver-
sions of gradient descent and Nes-
terov momentum compared on the
Rosenbrock function with b = 100;
see appendix B.6.

5.10 Summary

• Gradient descent follows the direction of steepest descent.

• The conjugate gradient method can automatically adjust to local valleys.

• Descent methods with momentum build up progress in favorable directions.

• A wide variety of accelerated descent methods use special techniques to speed
up descent.

• Hypergradient descent applies gradient descent to the learning rate of an
underlying descent method.

5.11 Exercises

Exercise 5.1. Compute the gradient of x⊤Ax + b⊤x when A is symmetric.
Exercise 5.2. Apply gradient descent with a unit step factor to f (x) = x4 from a
starting point of your choice. Compute two iterations.
Exercise 5.3. Apply one step of gradient descent to f (x) = ex + e−x from x(1) = 10

with both a unit step factor and with exact line search.
Exercise 5.4. The conjugate gradient method can also be used to find a search
direction d when a local quadratic model of a function is available at the current
point. With d as search direction, let the model be

q(d) = d⊤Hd + b⊤d + c

for a symmetric matrix H. What is the Hessian in this case? What is the gradient
of q when d = 0? What can go wrong if the conjugate gradient method is applied
to the quadratic model to get the search direction d?

Exercise 5.5. How is Nesterov momentum an improvement over momentum?

Exercise 5.6. In what way is the conjugate gradient method an improvement over
steepest descent?

Exercise 5.7. In conjugate gradient descent, what is the descent direction at
the first iteration for the function f (x, y) = x2 + xy + y2 + 5 when initialized
at (x, y) = (1, 1)? What is the resulting point after two steps of the conjugate
gradient method?

Exercise 5.8. We have a polynomial function f such that f (x) > 2 for all x in
three-dimensional Euclidean space. Suppose we are using steepest descent with
step lengths optimized at each step, and we want to find a local minimum of f .
If our unnormalized descent direction is [1, 2, 3] at step k, is it possible for our
unnormalized descent direction at step k + 1 to be [0, 0,−3]? Why or why not?

